
ВВЕДЕНИЕ 
 

Предлагаемый материал представляет лабораторные работы по теме 

«Макросы и их использование в пакете Microsoft Office» в рамках 

дисциплины «Технологии обработки информации». 

Цель преподавания дисциплины состоит в освоении студентами, 

получающими квалификацию бакалавра, принципов хранения и обработки 

на ЭВМ информации различных видов, технологий преобразования и 

передачи информации, а также подходов к автоматизации рутинных 

операций обработки информации различными программными средствами.  

Задачами дисциплины являются: развитие навыков работы с 

данными различной формы представления, понимание принципов 

кодирования, хранения, преобразования и передачи данных, освоение 

методов автоматизации обработки информации на ЭВМ. 

Макросом (или макрокомандой) называется последовательность 

заданных пользователем команд, имеющая имя и хранящаяся в форме 

стандартного программного модуля на языке программирования Visual 

Basic for Application (VBA). VBA – универсальный язык. Освоив его, 

можно получить ключ ко всем возможностям приложений Office. 

Предлагаемые лабораторные работы предназначены для развития у 

студентов устойчивых навыков в автоматизации повторяющихся действий 

и расширении функциональных возможностей приложений пакета 

Microsoft Office. 

 

  



5 
 

ЛАБОРАТОРНАЯ РАБОТА №1. Запись макросов с помощью 

макрорекордера 

 

Примечание: при записи макросов действия и параметры можно 

выбирать с помощью мыши, однако движения мыши не 

протоколируются. Например, мышь нельзя использовать для выделения, 

копирования, вставки и перетаскивания элементов. Для записи этих 

действий следует пользоваться клавиатурой. 

Задание 1. Работа со списком студентов: поменять местами 

фамилию и имя в каждом пункте. 

Работая в первой строке, необходимо произвести запись следующих 

действий с помощью макрорекордера (вкладка Разработчик): 

1) Выделить слово: Ctrl+Shift+→ 

2) Вырезать в буфер: Shift+Del 

3) Курсор в конец: End 

4) Вставить пробел 

5) Копировать из буфера Shift+Ins 

6) Курсор вниз к следующей строке ↓, затем в начало строки Home.  

 

Задание 2. Прочитать зашифрованный текст, применив к нему 

макрокоманду, программа которой приводится ниже. 

хеТлонигоо иарбтоб икфнимроицаи 

лоТокьас еымум ырди еас еымлг ыпун еп еддотюа ясубонеч 

юиноКцуфий 

 

1. Shift+→   7. Shift+← 

2. Shift+Del   8. Shift+Del 

3. →    9. ← 

4. →    10. Shift+Ins 

5. Shift+Ins   11. → 

6. ←    12. → 

 

Задание 3. Каждая строка документа содержит вопрос и ответ к нему 

(одно слово в конце строки).  Весь текст написан черным по белому, 

кеглем 14 пунктов: 

 

Требуется при помощи макрокоманды изменить формат ответов. 

Они должны быть написаны: 

1) кеглем 10 пунктов; 

2) курсивом; 

3) красными символами на белом фоне. 

Создать макрокоманду: 

1. Курсор в конец строки End; 

2. Выделить последнее слово Ctrl+Shift+←; 

3. Установить кегль в 10 пунктов; 

4. Установить курсивное начертание; 

5. Установить красный цвет; 

6. Курсор вниз к следующей строке ↓, в начало строки Home. 

7. Конец записи. 

Сколько бит в одном байте?    8 

Сколько байт в одном килобайте?   1024 

Сколько органов чувств у человека?  5 

Устройство для хранения информации?  Память 

Задание 4. Есть большой список слов, который необходимо 

преобразовать в один абзац: 

 

Действия для записи макроса: 

1) На первый символ следующего абзаца Ctrl+↓. 

2) Удалить символ конца абзаца Backspace. 

3) Вставить запятую и пробел. 

 

Задания для самостоятельной работы 

1. Создать макрос, с помощью которого слово, внутри которого 

находится курсор, подчеркивается и становится оформленным курсивом. 

2. В каждой строке записано одно слово (без лишних пробелов). 

Нужно поменять местами первую и последнюю буквы в каждом слове. 

 

 

ЛАБОРАТОРНАЯ РАБОТА №2. Кроссворд в MS Word 

 

Работа по созданию кроссворда 

1. Вставить в документ Word таблицу, в которой количество строк и 

столбцов соответствует максимальному количеству клеток 

подготовленного на бумаге кроссворда по горизонтали и по вертикали (в 

представленной версии 14 строк и 17 столбцов) (рис. 1). 

алгоритм 

алфавит 

байт 

бит 

браузер… 

алгоритм, алфавит, байт, бит, браузер,… 

2. Выделив эту таблицу, установить ширину столбцов и высоту 

строк таким образом, чтобы ячейки получились квадратными. 

3. Установить флажок Отобразить сетку (вкладка Макет). Убрать 

оформление ячеек таблицы командой Границы и заливка – Нет. 

4. Записать в виде макроса оформление блока ячеек по горизонтали: 

1) выделить блок ячеек под первое слово по горизонтали; 

2) записать макрос. Выполняемые действия: установка оформления 

для всех границ ячеек, затем заливки, размер шрифта 18, размещение 

текста по центру. 

5. Аналогично записать макрос оформления блока ячеек по 

вертикали. 

6. Для оформления кроссворда нужно выделять блоки ячеек и 

нажимать назначенную для макроса комбинацию клавиш. 

 
Рис. 1. Создание кроссворда в MS Word 

Затем необходимо проставить цифры в нужных ячейках. 

Макросы, автоматизирующие расстановку цифр по порядку для 

горизонтали и вертикали: 
Public n1, n2 As Integer 
Sub Номера1() 
n1 = n1 + 1 
Selection.Text = n1 & " ->" 
End Sub 
 
Sub Номера2() 
n2 = n2 + 1 
Selection.Text = n2 & " |" 
End Sub 
 
Sub Обнулить() 
n1 = 0 
n2 = 0 
End Sub 
 

Запись вопросов для созданного кроссворда 

Макрос, выдающий диалоговое окно с вопросом, в зависимости от 

выделенной ячейки с номером вопроса: 
Sub Вопросы() 

Select Case Selection 

Case "1 ->" 

MsgBox "Средство хранения информации." 

Case "2 ->" 

MsgBox "И судебный, и сетевой." 

Case "3 ->" 

MsgBox "Он занимает объем памяти 1 байт." 

Case "4 ->" 

MsgBox "Место визуализации электронного документа." 

Case "5 ->" 

MsgBox "По нему передвигают мышь." 

Case "6 ->" 

MsgBox "Множество точек в графическом изображении." 

Case "7 ->" 

MsgBox "Тип данных в Excel." 

Case "8 ->" 

MsgBox "Поименованная область на диске." 

Case "9 ->" 

MsgBox "Точка на экране." 

Case "10 ->" 

MsgBox "Основной элемент электронной таблицы." 

Case "1 |" 

MsgBox "Графопостроитель." 

Case "2 |" 

MsgBox "Процедура проверки." 

Case "3 |" 

MsgBox "Запись символа в памяти." 

Case "4 |" 

MsgBox "Наука о способах хранения, передачи и обработки 

информации." 

Case "5 |" 

MsgBox "Стандартный графический объект в Windows." 

Case "6 |" 

MsgBox "...команда." 

Case "7 |" 

MsgBox "По-русски - отрицание, по-английски - сеть." 

End Select 

End Sub 

 

 

ЛАБОРАТОРНАЯ РАБОТА №3. Автоматизация формирования 

приказа о выплате премии 

 

1. Создать новый макрос с именем «Премия». 

2. В редакторе VBA добавить форму Insert – UserForm. 

3. В созданный макрос записать код: 
Sub Премия() 
UserForm1.Show 
End Sub 
 

4. Оформить внешний вид формы с использованием Label, TextBox, 

OptionButton, CheckBox, а также кнопок CommandButton (рис. 2). 

 
Рис. 2. Форма приказа о выплате премии 

5. Первое текстовое поле появляется только при выборе пункта 

«другое» (изначально свойство Visible для TextBox1 устанавливается 

False): 
Private Sub OptionButton3_Click() 
TextBox1.Visible = True 
End Sub 
 
6. Аналогично текстовое поле для ввода суммы появляется только 

при установке флажка CheckBox1: 
Private Sub CheckBox1_Click() 
If CheckBox1.Value = True Then 
Label3.Visible = True 



12 
 

TextBox3.Visible = True 
Label4.Visible = True 
Else 
Label3.Visible = False 
TextBox3.Visible = False 
Label4.Visible = False 
End If 
End Sub 
 

7. Кнопка «Напечатать приказ» собирает данные, введенные в 

форму, и формирует текст документа: 
Private Sub CommandButton2_Click() 
Dim Povod, Fio As String 
Dim Summa As Integer 
If OptionButton1.Value = True Then Povod = OptionButton1.Caption 
If OptionButton2.Value = True Then Povod = OptionButton2.Caption 
If OptionButton3.Value = True Then Povod = TextBox1.Text 
Fio = TextBox2.Text 
Summa = Val(TextBox3.Text) 
With Selection 
    .Style = "Заголовок 1" 
    .ParagraphFormat.Alignment = wdAlignParagraphCenter 
    .TypeText "Приказ" 
    .TypeText vbCrLf 
    .TypeText vbCrLf 
    .Style = "Обычный" 
    .ParagraphFormat.Alignment = wdAlignParagraphCenter 
    .Font.Name = "Times New Roman" 
    .Font.Size = 14 
    .TypeText "г.Грозный" & Space(70) & Date 
    .TypeText vbCrLf 
    .TypeText vbCrLf 
    .ParagraphFormat.Alignment = wdAlignParagraphJustify 
    .TypeText vbTab 
    .TypeText "За проявленные успехи в " & Povod & " наградить " & Fio & ":" 
    .TypeText vbCrLf 
If CheckBox1.Value = True Then 
    .TypeText vbTab & "- денежной премией в сумме " & Summa & " руб." 
    .TypeText vbCrLf 
End If 
If CheckBox2.Value = True Then 
    .TypeText vbTab & "- почетной грамотой." 
Else 

    .TypeText "." 
End If 
    .TypeText vbCrLf 
    .TypeText vbCrLf 
    .ParagraphFormat.Alignment = wdAlignParagraphCenter 
    .TypeText "Генеральный директор" & Space(50) & "Ахмадов А.А." 
    .TypeText vbCrLf 
End With 
End 
End Sub 
 
8. Кнопка «Отмена» закрывает форму: 
Private Sub CommandButton1_Click() 
End 
End Sub 
 

Задание для самостоятельной работы 

Разработать программу, позволяющую заполнять объяснительную 

записку о причине пропусков занятий (семейные обстоятельства, плохое 

состояние здоровья и т.д.). Причину пропуска предусмотреть в 

раскрывающемся списке создаваемой формы. 

 

 

ЛАБОРАТОРНАЯ РАБОТА №4. Создание макроса для ведения 

учета ключей на кафедре 

 

Макрос позволяет заполнять данными форму следующего вида (рис. 

3). 

Рис. 3. Форма для ведения учета ключей 

В результате формируется таблица с записями (рис. 4). 

 
Рис. 4. Таблица со сведениями по учету ключей 

1. Создать новый макрос с именем «Выдача ключей»: 
Sub Выдача_ключей() 
UserForm2.Show 
End Sub 
 

2. Кнопка «Создать таблицу» формирует основу таблицы из 6-и 

столбцов с заданной шапкой: 
Public Table1 As Table 
Private Sub CommandButton1_Click() 
Selection.Font.Name = "Times New Roman" 
Selection.Font.Size = 12 
 
Set Table1 = ThisDocument.Tables.Add(Selection.Range, 1, 6) 
n = 1 
Table1.Style = "Сетка таблицы" 
Table1.Columns(1).Width = 25 
Table1.Columns(2).Width = 100 
Table1.Columns(3).Width = 50 
Table1.Columns(4).Width = 200 
Table1.Columns(5).Width = 100 
Table1.Columns(6).Width = 200 

Table1.Cell(1, 1).Range.Text = "№" 
Table1.Cell(1, 2).Range.Text = "дата" 
Table1.Cell(1, 3).Range.Text = "ауд." 
Table1.Cell(1, 4).Range.Text = "сотрудник" 
Table1.Cell(1, 5).Range.Text = "взял/вернул" 
Table1.Cell(1, 6).Range.Text = "отв.дежурный" 
UserForm2.Hide 
End Sub 
 

3. Кнопка «Добавить запись» заполняет строку таблицы: 
Private Sub CommandButton2_Click() 
Dim n As Integer 
Set Table1 = ThisDocument.Tables(ThisDocument.Tables.Count) 
n = Table1.Rows.Count 
Table1.Rows(n).Select 
Selection.InsertRowsBelow 1 
 
Table1.Cell(n + 1, 1).Range.Text = n 
Table1.Cell(n + 1, 2).Range.Text = Date & ", " & Time 
Table1.Cell(n + 1, 3).Range.Text = ComboBox1.Value 
Table1.Cell(n + 1, 4).Range.Text = TextBox1.Text 
If OptionButton1.Value = True Then 
Table1.Cell(n + 1, 5).Range.Text = "взял" 
Else 
Table1.Cell(n + 1, 5).Range.Text = "вернул" 
End If 
Table1.Cell(n + 1, 6).Range.Text = TextBox2.Text 
UserForm2.Hide 
End Sub 

 
Private Sub UserForm_Initialize() 
ComboBox1.AddItem "4-01" 
ComboBox1.AddItem "4-02" 

ЛАБОРАТОРНАЯ РАБОТА №5. Обработка текста с 

определенным форматированием 

 

Обработать текст: все предложения, в которых есть выделение слов 

определенным цветом (рис. 5), скопировать в отдельный документ. 

Например, текст исходного документа: 

Используя макросы, можно значительно (в десятки, а то и в сотни 

раз) увеличить эффективность работы с приложением. Помимо 

выполнения набора встроенных в приложение команд, макрос позволяет 

производить обработку внешних файлов, загрузку и передачу файлов и 

данных через интернет, чтение и изменение настроек операционной 

системы. В большинстве случаев макрорекордер (режим записи действий 

пользователя в виде макроса) выдаёт код, нуждающийся в доработке. Но, 

тем не менее, и такой макрос значительно сокращает время, требующееся 

на выполнение рутинных операций. 

Корректный перевод термина с английского – «макрокоманда», 

слово же «макрос» получило распространение благодаря использованию в 

локализованных продуктах американской корпорации Microsoft. 

 
Рис. 5. Выделение текста цветом в MS Word 

Результатом работы макроса является новый документ с текстом: 

Используя макросы, можно значительно (в десятки, а то и в сотни 

раз) увеличить эффективность работы с приложением. 

 Корректный перевод термина с английского — «макрокоманда», 

слово же «макрос» получило распространение благодаря использованию в 

локализованных продуктах американской корпорации Microsoft. 

Макрос: 
Sub Выделенный_текст() 
Dim longDoc, shortDoc As Document 
Dim myword, mysentence As Range 
 
Set longDoc = ThisDocument 
Set shortDoc = Documents.Add() 
shortDoc.SaveAs (longDoc.Path + "\(short)" + longDoc.Name) 
 
For Each mysentence In ThisDocument.Sentences 
For Each myword In mysentence.Words 
If myword.HighlightColorIndex = Options.DefaultHighlightColorIndex Then 
mysentence.Copy 
shortDoc.Activate 
Selection.TypeText Text:=" " 
Selection.PasteAndFormat (wdPasteDefault) 
Selection.TypeParagraph 
longDoc.Activate 
Exit For 
End If 
Next 
Next 
End Sub 

 

 

ЛАБОРАТОРНАЯ РАБОТА №6. Использование закладок 

 

В отдельном файле хранятся часто используемые при составлении 

документов фрагменты (например, официальное наименование 

организации, подписные листы, варианты бланков). 

Написать макрос, позволяющий легко выбирать и вставлять эти 

фрагменты в составляемый документ. 

Материал для документа «Фрагменты.docx» (файл предоставляется 

студентам): 

 

Генеральный директор  ______________           Фамилия И.О. 

 

Начальник юридического отдела  Фамилия И.О. 

Начальник отдела кадров  Фамилия И.О. 

Начальник общего отдела   Фамилия И.О. 

 

Грозненский Государственный Нефтяной Технический Университет 

им. акад. М.Д. Миллионщикова 

ОАО  «…» 

адрес:  г.Грозный, … 

тел.:  8-(928)-…-...-… 

Бланк № 

     

  

  

     

 

 

Бланк №   

   

   

 

Выбираются фрагменты из списка (рис. 6): 

Рис. 6. Список фрагментов 

Каждый фрагмент необходимо сделать закладкой (Bookmark). Для 

этого фрагмент текста (или рисунок / таблица) выделяется, и командой 

меню Вставка – Закладка объявляется закладкой. 

В приведенном примере 6 различных блоков-закладок с именами от 

z1 до z6. 

Код VBA для кнопки «Вставить»: 
Private Sub CommandButton1_Click() 
Dim newdoc, xdoc As Document 
Set newdoc = ActiveDocument 
Set xdoc = Documents.Open(newdoc.Path + "\Фрагменты2.docx") 
If ComboBox1.Value = "ген.директор" Then 
xdoc.Bookmarks("z1").Range.Copy 
If ComboBox1.Value = "отделы" Then xdoc.Bookmarks("z2").Range.Copy 
If ComboBox1.Value = "наименование организации" Then 
xdoc.Bookmarks("z3").Range.Copy 
If ComboBox1.Value = "представление организации" Then 
xdoc.Bookmarks("z4").Range.Copy 
If ComboBox1.Value = "таблица-бланк 1" Then 
xdoc.Bookmarks("z5").Range.Copy 

If ComboBox1.Value = "таблица-бланк 2" Then 
xdoc.Bookmarks("z6").Range.Copy 
 
newdoc.Activate 
Selection.PasteAndFormat (wdPasteDefault) 
End Sub 
 

Код для заполнения списка наименованиями фрагментов: 
Private Sub UserForm_Initialize() 
ComboBox1.AddItem "ген.директор" 
ComboBox1.AddItem "отделы" 
ComboBox1.AddItem "наименование организации" 
ComboBox1.AddItem "представление организации" 
ComboBox1.AddItem "таблица-бланк 1" 
ComboBox1.AddItem "таблица-бланк 2" 
End Sub 
 

 

ЛАБОРАТОРНАЯ РАБОТА №7. Работа с абзацами, 

предложениями и словами 

 

Разработать макрос, реализующий следующие действия (рис. 7): 

 
Рис. 7. Команды для обработки текста 

Выделение красным цветом первого слова каждого абзаца в тексте: 

Private Sub CommandButton1_Click() 
Dim doc As Document 
Dim paragr As Paragraph 
 
Set doc = ActiveDocument 
For Each paragr In doc.Paragraphs 
paragr.Range.Words(1).Font.Color = wdColorRed 
Next 
UserForm1.Hide 
End Sub 
 

Перевод в верхний регистр выделенного слова: 
Private Sub CommandButton2_Click() 
Dim doc As Document 
Dim s As String 
 
Set doc = ActiveDocument 
Selection.Expand unit:=wdWord 
s = Selection.Text 
s = UCase(s) 
Selection.TypeText s 
UserForm1.Hide 
End Sub 
 
Подчеркивание предложения, в котором установлен курсор ввода: 
Private Sub CommandButton3_Click() 
Dim doc As Document 
 
Set doc = ActiveDocument 
Selection.Expand unit:=wdSentence 
Selection.Font.Underline = wdUnderlineSingle 
UserForm1.Hide 
End Sub 
 

 

ЛАБОРАТОРНАЯ РАБОТА №8. Работа с таблицами в MS Excel 

 

1. Создать в файле Excel таблицу (рис. 8). 

Рис. 8. Таблица клиентов 

Лист 1 переименовать в «Клиенты». 

2. Перейдя на вкладку Разработчик, вставить элемент ActiveX – 

кнопку. Свойство Caption – «Ввести нового клиента». 

В режиме конструктора двойной щелчок по кнопке открывает 

редактор программного кода. Процедура для кнопки: 
Private Sub CommandButton1_Click() 
UserForm1.Show 
End Sub 
 

3. Пользовательская форма добавляется командой Insert – UserForm 

(рис. 9). 

 
Рис. 9. Форма для добавления клиента 

В свойстве Name для удобства написания кода сменить имена 

объектов формы: текстовые поля Cod, Firma, Predst, Adres, Telefon, INN, 

кнопка btnClient. 

4. Код для кнопки на UserForm: 
Private Sub btnClient_Click() 
Dim Nstroki As Integer 
Dim i As Integer 
If Cod.Text = "" Then 
MsgBox ("Необходимо указать код организации!") 
Exit Sub 
End If 
 
Nstroki = 2 
Do While Worksheets("Клиенты").Cells(Nstroki, 2).Value <> "" 
Nstroki = Nstroki + 1 
Loop 
 
For i = 2 To Nstroki - 1 
If Worksheets("Клиенты").Cells(i, 2).Value = Firma.Text _ 
And Worksheets("Клиенты").Cells(i, 4).Value = Adres.Text Then 
MsgBox ("Такая фирма уже есть в списке!") 
Exit Sub 
End If 
If Worksheets("Клиенты").Cells(i, 1).Value = Cod.Text Then 
MsgBox ("Данный код уже использован!") 
Exit Sub 
End If 
Next 
 
Worksheets("Клиенты").Cells(Nstroki, 1).Value = Cod.Text 
Worksheets("Клиенты").Cells(Nstroki, 2).Value = Firma.Text 
Worksheets("Клиенты").Cells(Nstroki, 3).Value = Predst.Text 
Worksheets("Клиенты").Cells(Nstroki, 4).Value = Adres.Text 
Worksheets("Клиенты").Cells(Nstroki, 5).Value = Telefon.Text 
Worksheets("Клиенты").Cells(Nstroki, 6).Value = INN.Text 
Worksheets("Клиенты").UsedRange.Borders.LineStyle = xlContinuous 
End Sub 

ЛАБОРАТОРНАЯ РАБОТА №9. Учет продаж 

 

1. На первом листе рабочей книги Excel создать список 

продаваемых товаров (рис. 10). 

 
Рис. 10. Список товаров 

Лист назвать «Товары». 

2. Оформить второй лист: 

заполнить шапку таблицы (наименование, количество, сумма); 

разместить элементы ActiveX: две кнопки («Добавить в заказ» и 

«Сброс»), поле со списком (ComboBox) и подпись (Label) (рис. 11). 

 
Рис. 11. Оформление заказа 

Второй лист назвать «Заказ». 

3. ComboBox должен заполняться автоматически при открытии 

рабочей книги.  

Для этого необходимо создать процедуру Workbook_Open: нужно в 

редакторе VBA дважды щелкнуть на пиктограмме «Эта книга» (рис. 12). 

 
Рис. 12. Окно Project в редакторе VBA 

Код процедуры Workbook_Open: 
Private Sub Workbook_Open() 
Dim k, i As Integer 
Dim tovar As String 
 
Worksheets("Заказ").ComboBox1.Clear 
Worksheets("Заказ").Label1.Caption = "0" 
 
k = Worksheets("Товары").UsedRange.Rows.Count 
For i = 1 To k 
tovar = Worksheets("Товары").Cells(i, 1).Value & " (" &       

Worksheets("Товары").Cells(i, 2).Value & ")" 
Worksheets("Заказ").ComboBox1.AddItem tovar 
Next 
End Sub 
 
4. Код для кнопки «Добавить в заказ»: 
Private Sub CommandButton1_Click() 
Dim NStroki, kol, cena As Integer 
NStroki = Worksheets ("Заказ").UsedRange.Rows.Count + 1 
 
Worksheets ("Заказ").Cells (NStroki, 1).Value = ComboBox1.Value 
kol = InputBox ("Введите количество:", "Оформление заказа", 1) 

Worksheets ("Заказ").Cells(NStroki, 2).Value = kol 
cena = Worksheets ("Товары").Cells(ComboBox1.ListIndex + 1, 2) 
Worksheets ("Заказ").Cells (NStroki, 3).Value = kol * cena 
 
Label1.Caption = Val (Label1.Caption) + kol * cena 
End Sub 
 

5. Код для кнопки «Сброс»: 
Private Sub CommandButton2_Click() 
Label1.Caption = "0" 
End Sub 
 

Лист «Товары» можно скрыть от пользователя командой 

Worksheets("Товары").Visible = False 

в процедуре Workbook_Open. 

 

ЛАБОРАТОРНАЯ РАБОТА №10. «Игра на матрице» в MS Excel 

 

Игра происходит на поле 5х5 клеток. Два игрока, условно 

называемые «четный» и «нечетный», поочередно ставят нули и единицы в 

свободные клетки поля. Каждый из игроков может ставить 0 или 1 в 

произвольную свободную клетку, тем самым занимая ее. Игра 

продолжается до заполнения всех клеток на поле.  

После этого отдельно суммируются числа вдоль каждой строки, 

каждого столбца и вдоль двух диагоналей. В поле 5х5 таких сумм будет 12 

(5 по строкам, 5 по столбцам и 2 по диагоналям). Если количество четных 

сумм получится больше, то выигрывает «четный» игрок, а если наоборот, 

то «нечетный». 

Интерфейс игры – рисунок 13.  

Рис. 13. «Игра на матрице» 

Поле 5х5 клеток оформляется вручную: границы, заливка, шрифт. 

Дополнительными компонентами являются одна кнопка и два Label 

для вывода количества сумм.  

Технически ходы выполняются двойным щелчком по нужной клетке. 

Если клетка пуста или содержит 0, в нее ставится 1; повторный двойной 

щелчок меняет значение обратно на 0. 

Двойной щелчок на листе обрабатывает процедура 

Worksheet_BeforeDoubleClick (рис. 14). 

 
Рис. 14. Процедура Worksheet_BeforeDoubleClick 

Код обработки двойного щелчка: 
Private Sub Worksheet_BeforeDoubleClick(ByVal Target As Range, Cancel 

As Boolean) 

Dim Sum(12) As Integer    'Массив для хранения сумм 
Dim i, j As Integer 
Dim k, n As Integer   'Колич-во четных и нечетных сумм 
 
'Смена значения в ячейке, на которой произведен двойной щелчок 
If ActiveCell.Value = 0 Then 
ActiveCell.Value = 1 
Else 
ActiveCell.Value = 0 
End If 
'Вычисление сумм по строкам 
For i = 1 To 5 
Sum(i) = 0 
For j = 1 To 5 
Sum(i) = Sum(i) + Cells(i, j).Value 
Next 
Next 
'Вычисление сумм по столбцам 
For i = 1 To 5 
Sum(i + 5) = 0 
For j = 1 To 5 
Sum(i + 5) = Sum(i + 5) + Cells(j, i).Value 
Next 
Next 
'Вычисление сумм по диагоналям  
Sum(11) = 0 
For i = 1 To 5 
Sum(11) = Sum(11) + Cells(i, i).Value 
Next 
Sum(12) = 0 
For i = 1 To 5 
Sum(12) = Sum(12) + Cells(6 - i, i).Value 
Next 
'Подчет четных и нечетных сумм 
k = 0 
n = 0 
For i = 1 To 12 
If Sum(i) Mod 2 = 0 Then 
k = k + 1 
Else 
n = n + 1 
End If 

Next 
Label1.Caption = k 
Label2.Caption = n 
Cells(7, 7).Activate 
End Sub 
 

Код для кнопки «Начать игру»: 
Private Sub CommandButton1_Click() 
Label1.Caption = "0" 
Label2.Caption = "0" 
Range("A1:E5").Select 
Selection.Value = "" 
End Sub 
 

 

ЛАБОРАТОРНАЯ РАБОТА №11. Имитация движения 

шахматного коня 

 

Подготовка интерфейса – рисунок 15. 

Размещаемые компоненты: два текстовых поля для ввода начальных 

координат шахматной фигуры, одно текстовое поле для вывода количества 

ходов; четыре кнопки. 

Кнопки предлагается переименовать: 

«Старт» - btnStart, 

«Варианты хода» - btnVar, 

«Сделать ход» - btnDo, 

«Очистить доску» - btnClr. 

Рис. 15. «Шахматная доска» в MS Excel 

При нажатии кнопки Старт «конь» занимает указанную начальную 

позицию на поле. Кнопка Варианты хода одну за другой предлагает новые 

возможные позиции для перехода, выделяя ячейки цветом. 

Существует ограниченный набор ячеек, куда может попасть конь из 

любой своей позиции. Максимальное число вариантов – 8.  

Коды: 
Dim x, y, newx, newy As Integer 
Dim k As Integer     'Номер варианта хода 
Dim N As Integer     'Общее количество ходов 
 
Private Sub btnStart_Click() 
x = Val(TextBox1.Text) 
y = Val(TextBox2.Text) 
Cells(x, y).Value = "X" 
Cells(x, y).Interior.ColorIndex = 4 
k = 0 
N = 0 
End Sub 
 
Private Sub btnVar_Click() 
Dim masx As Variant 
Dim masy As Variant 
'Возможные смещения коня 

masx = Array(1, 2, 2, 1, -1, -2, -2, -1) 
masy = Array(-2, -1, 1, 2, 2, 1, -1, -2) 
 
'Выбор белого цвета заливки всего поля 
Range("A1:H8").Interior.ColorIndex = 0 
 
newx = x + masx(k) 
newy = y + masy(k) 
If k = 7 Then 
k = 0 
Else 
k = k + 1 
End If 
 
'Выход за границы поля не допускается 
If newx < 1 Or newx > 8 Or newy < 1 Or newy > 8 Then 
Exit Sub 
End If 
'Переход на уже пройденную клетку невозможен 
If Cells(newx, newy).Value <> "X" Then 
Cells(newx, newy).Interior.ColorIndex = 5 
End If 
 
End Sub 
 
Private Sub btnDo_Click() 
x = newx 
y = newy 
Cells(newx, newy).Value = "X" 
Cells(newx, newy).Interior.ColorIndex = 4 
k = 0 
N = N + 1 
TextBox3.Text = N 
End Sub 
 
Private Sub btnClr_Click() 
Range("A1:H8").Interior.ColorIndex = 0 
Range("A1:H8").Value = "" 
N = 0 
TextBox3.Text = N 
End Sub 

ЛАБОРАТОРНАЯ РАБОТА №12. Игра с заполнением полей 

 

Каждому из двух игроков предоставляется для совершения хода одно 

из двух направлений – горизонтальное или вертикальное. Один игрок 

всегда ходит только по горизонтали, второй – только по вертикали. Ходы 

делаются поочередно. Переходить можно только в незанятую клетку. 

При этом «горизонтальный» игрок, кроме клетки, которую он занял 

сам, автоматически захватывает и все соседние свободные в этой строке 

(клетки закрашиваются и получают знак «0»). Аналогично «вертикальный» 

игрок занимает все рядом стоящие свободные клетки в выбранном столбце 

(знак «Х») (рис. 16).  

Выбор клетки для перехода осуществляется с помощью полос 

прокрутки. Их свойства:  

Name – Horizontal, Orientation – frmOrientationHorizontal, Min – 1, 

Max – 5 (горизонтальная полоса прокрутки);  

Name – Vertical, Orientation – frmOrientationVertical, Min – 1, Max – 5 

(вертикальная). 

 
Рис. 16. Интерфейс игры с заполнением полей 

Первый ход делается игроком по горизонтали, задача второго игрока 

– не давать «горизонтальному» игроку заполнить все поле.  

Программный код: 
Private Sub CommandButton1_Click()    'Кнопка "По горизонтали" 
Dim x, y As Integer 
Dim i As Integer 
x = Vertical.Value 
y = Horizontal.Value 
If Cells(x, y).Interior.ColorIndex = 6 Then 
MsgBox ("Ход невозможен") 
Else 
'Заливка выбранной ячейки 
Cells(x, y).Interior.ColorIndex = 6 
Cells(x, y).Value = "0" 
'Заливка ячеек, расположенных справа 
For i = y + 1 To 5 
If Cells(x, i).Interior.ColorIndex <> 6 Then 
Cells(x, i).Interior.ColorIndex = 6 
Cells(x, i).Value = "0" 
Else 
Exit For 
End If 
Next 
'Заливка ячеек, расположенных слева от выбранной 
For i = y - 1 To 1 Step -1 
If Cells(x, i).Interior.ColorIndex <> 6 Then 
Cells(x, i).Interior.ColorIndex = 6 
Cells(x, i).Value = "0" 
Else 
Exit For 
End If 
Next 
 
CommandButton2.Enabled = True 
CommandButton1.Enabled = False 
End If 
End Sub 
 
Private Sub CommandButton2_Click()  'Кнопка "По вертикали" 
Dim x, y As Integer 
Dim i As Integer 

x = Vertical.Value 
y = Horizontal.Value 
If Cells(x, y).Interior.ColorIndex = 6 Then 
MsgBox ("Ход невозможен") 
Else 
'Заливка выбранной ячейки 
Cells(x, y).Interior.ColorIndex = 6 
Cells(x, y).Value = "X" 
'Заливка ячеек, расположенных справа 
For i = x + 1 To 5 
If Cells(i, y).Interior.ColorIndex <> 6 Then 
Cells(i, y).Interior.ColorIndex = 6 
Cells(i, y).Value = "X" 
Else 
Exit For 
End If 
Next 
'Заливка ячеек, расположенных слева от выбранной 
For i = x - 1 To 1 Step -1 
If Cells(i, y).Interior.ColorIndex <> 6 Then 
Cells(i, y).Interior.ColorIndex = 6 
Cells(i, y).Value = "X" 
Else 
Exit For 
End If 
Next 
 
CommandButton1.Enabled = True 
CommandButton2.Enabled = False 
End If 
End Sub 
 
Private Sub CommandButton3_Click()  'Кнопка "Очистить" 
Range("A1:E5").Interior.ColorIndex = 0 
Range("A1:E5").Value = "" 
CommandButton1.Enabled = True 
CommandButton2.Enabled = False 
End Sub 

ЛАБОРАТОРНАЯ РАБОТА №13. Макросы в PowerPoint 

 

Требуется автоматически создать презентацию на основе картинок, 

содержащихся в отдельном каталоге. Файлов в каталоге – произвольное 

количество. Необходимо разместить картинки по слайдам, выровняв 

размер картинок.  

1. С вкладки «Разработчик» создается макрос (рис. 17). 

 
Рис. 17. Создание макроса в MS Power Point 

2. Для удобного обращения к объектам файловой системы 

подключается дополнительная библиотека: меню Tools – References – 

пункт Microsoft Scripting Runtime (рис. 18). 

 
Рис. 18. Подключение библиотеки 

3. Код макроса: 
Sub Картинки() 
Dim present As Presentation 
Dim xslide As Slide 
Dim fso As New Scripting.FileSystemObject 
Dim xfolder As Scripting.Folder 
Dim xfile As Scripting.File 
Dim path As String 
Dim i As Integer 
 
Set present = ActivePresentation 
path = InputBox("Введите путь каталога:", "Папка с картинками") 
Set xfolder = fso.GetFolder(path) 
 
i = 1 
For Each xfile In xfolder.Files 
Set xslide = present.Slides.Add(i, ppLayoutBlank) 
xslide.Shapes.AddPicture FileName:=path & "\" & xfile.Name, 
LinkToFile:=msoFalse, _ 
SaveWithDocument:=msoTrue, Left:=10, Top:=10, Width:=700, Height:=520 
i = i + 1 
Next 
End Sub 
 

Путь к каталогу с картинками запрашивается у пользователя с 

помощью окна ввода InputBox. В указанном каталоге с помощью цикла 

перебираются все файлы.  

